# *Nonlinear Differential Equations: Pendulum*

In this notebook, we solve nonlinear differential equations, looking at the time dependence, phase space plots, Poincare sections, and the power spectrum.

The notebook has specialized to a pendulum equation, with parameters used in session 8 of 780.20.

## Define the Differential Equation

Name the equation "diffeq". Note the "==" in defining the equation.

```
diffeq := θ''[t] + α θ'[t] + ω0^2 Sin[θ[t]] == fext Cos[ωext * t + ϕext]
```

Choose among the parameters:

```
p1 := {α = 0., ω0 = 1, fext = 0., ωext = 0.689, ϕext = 0.0};

p2 := {α = 0.2, ω0 = 1, fext = 0.52, ωext = 0.689, ϕext = 0.0};

p3 := {α = 0.2, ω0 = 1, fext = 0.52, ωext = 0.694, ϕext = 0.0};

p4 := {α = 0.2, ω0 = 1, fext = 0.52, ωext = 0.689, ϕext = 0.0};

p5 := {α = 0.2, ω0 = 1, fext = 0.9, ωext = 0.54, ϕext = 0.0};

p2
```

```
{0.2, 1, 0.52, 0.689, 0.}
```

Choose among the initial conditions:

```
ic1 := {θ0 = 0.8, θdot0 = 0.};

ic2 := {θ0 = -0.8, θdot0 = 0.1234};

ic3 := {θ0 = 0.8, θdot0 = 0.8};

ic4 := {θ0 = 0.8, θdot0 = 0.8};

ic5 := {θ0 = -0.8, θdot0 = 0.1234};

ic2
```

```
{-0.8, 0.1234}
```

## Solve the Differential Equation

Specify the range in time over which we will solve the differential equation. We won't be able to use the solution outside of this range. (I.e., we'll have to extend this range if we get an "outside of range" error later.)

```
tmin = 0; tmax = 1000;
```

Numerically solve the differential equaiton using NDSolve, specifying the initial conditions. Setting MaxSteps to a large number is needed if tmax is large.

```
solution = NDSolve[{diffeq, θ[0] == θ0, θ'[0] == θdot0},
   θ, {t, tmin, tmax}, MaxSteps → 20 000];
```

NDSolve returns an "interpolating function", which can be evaluated later at any time t in tmin < t < tmax.

## Plot the Time Dependence and Phase Space

We can just use "Plot" with Evaluate and the "interpolating function" defined by "solution".

```
Plot[Evaluate[θ[t] /. solution], {t, 0, 100}]
```




Or we can do a parametric plot ("ParametricPlot") with the same solution. We've added various options here to (try to) make a nicer plot.

```
ParametricPlot[Evaluate[{t, θ[t]} /. solution],
 {t, 0, 100}, PlotStyle → Hue[0.6`],
 BaseStyle → {FontFamily → "Times", FontSize → 14},
 ImageSize → {350, 200}, PlotLabel → Style["Time Dependence"],
 AxesLabel → {"t", "θ"}, AspectRatio → 1 / 2]
```




Now do the phase space plot. You may need to change AxesOrigin to get the axes in a reasonable place.

```
ParametricPlot[Evaluate[{θ[t], θ'[t]} /. solution],
 {t, tmin, 100}, PlotStyle → Hue[0.6`], AspectRatio → 1,
 BaseStyle → {FontFamily → "Times", FontSize → 14},
 ImageSize → {250, 250}, PlotLabel → Style["Phase Space"],
 AxesOrigin → {-1, -1}, AxesLabel → {"θ", "θ'"}]
```




Here is the same plot, with the starting time t set to skip the transient region.

```
ParametricPlot[Evaluate[{θ[t], θ'[t]} /. solution],
 {t, 100, 300}, PlotStyle → Hue[0.6`], AspectRatio → 1,
 BaseStyle → {FontFamily → "Times", FontSize → 14},
 ImageSize → {250, 250}, PlotLabel → Style["Phase Space"],
 AxesOrigin → {-1, -1}, AxesLabel → {"θ", "θ'"}]
```




## Poincare Sections

The idea of a Poincare section is to plot a point in phase space once every period of the external force, $2\pi$/(external frequency). The resulting pattern gives information about the periodicity of the signal (or indicates chaos). Start the plot at a large enough time t ("tstart") so that the transients have died out.

Set the external period and how many periods we'll consider. Define a function timeperiod[i] giving the corresponding time as a function of the period number.

```
Texternal := 2 Pi / ωext;

tstart = 10 * Texternal;

numperiods = 20;

timeperiod[i_] := tstart + i * Texternal
```

Just evaluate at the relevant points. Flatten[expr,1] strips off a layer of {}'s.

```
PoincarePts = Flatten[
   Table[Evaluate[{θ[timeperiod[i]], θ'[timeperiod[i]]} /. solution],
    {i, 0, numperiods}], 1];
```

ListPlot plots pairs of numbers in the PlotRange.

```
ListPlot[PoincarePts, AspectRatio → 1,
 BaseStyle → {FontFamily → "Times", FontSize → 14},
 ImageSize → {300, 300}, PlotRange → {{-π, π}, {-2, 2}},
 PlotLabel → Style["Poincare Section"],
 AxesOrigin → {-π, 0}, AxesLabel → {"θ", "θ'"},
 PlotStyle → {PointSize[0.015`], RGBColor[1, 0, 0]}]
```




## Power Spectrum

The power spectrum is found by taking a Fourier transform (FFT) of the signal. We generate points at the same time intervals as in diffeq_pendulum.cpp.

```
Tskip = 100;

time[i_] := tstart + i * Texternal / Tskip

numpoints = 5000;

SampleTime := numpoints * Texternal / Tskip

SamplePts = Flatten[
   Table[Evaluate[θ[time[i]] /. solution], {i, 0, numpoints}], 1];
```

Check first that the plot still looks ok.

```
ListPlot[SamplePts, PlotRange → {{0, 1000}, Automatic}]
```




The basic command is Fourier. FourierParameters chooses a particular conention for defining the transform. Abs takes the magnitude of the resulting complex number (which we then square).

```
transform = Abs[Fourier[SamplePts, FourierParameters → {-1, 1}]]^2;
```

```
frequencies = Table[(2 Pi / SampleTime) * i, {i, 0, numpoints}];
```

Transpose to make pairs of points from two separate lists of points.

```
PlotPts = Transpose[{frequencies, transform}];
```

```
ListLogPlot[PlotPts, Joined → True, PlotRange → {{0, 2}, {0.001`, 2}},
 BaseStyle → {FontFamily → "Times", FontSize → 14},
 ImageSize → {400, 400}, PlotLabel → Style["Power Spectrum"],
 AxesOrigin → Automatic, AxesLabel → {"freq.", "power"}]
```